
TTCN‑3 supports the union type. The union type is a collection of alternatives, each one identified by an identifier. Only one of the specified alternatives will ever be present in an actual union value. Union types are useful to model data which can take one of a finite number of known types.
EXAMPLE 1:
	type union MyUnionType 
	{	
		integer 		number,
		charstring 		string
	};

	// A valid instantiation of MyUnionType would be 
	var MyUnionType v_age, v_oneYearOlder;
	var integer v_ageInMonths;

	v_age.number := 34; 		// value notation by referencing the field. Note, that this 
							    // notation makes the given field to be the chosen one
	v_oneYearOlder := {number := v_age.number+1};

	v_ageInMonths := v_age.number * 12;

The assignment notation shall be used for union-s, and the notation shall assign a value to one field only. This field becomes the chosen field. Neither the not used symbol "-" nor omit is allowed in union value notations.
The value list notation shall not be used for setting values of union types.
At most one of the union alternatives can be declared as the default alternative by using the @default modifier before the type of the alternative. For unions with a default alternative, special type compatibility rules apply (see clause 6.3.2.4) which allow using the union value as compatible with the type of the default alternative. Therefore, the assignment notation does not have to be used to denote a value of the union type if the union's default alternative is to be chosen. Also, the default alternative selection does not have to be used to access the default alternative, if it is chosen.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a) If the default alternative is of a record or set type, identifiers of fields of the record or set type and identifiers of  alternatives of the union type containing the default alternative shall be distinct.

b) If the default alternative is of a union type, identifiers of alternatives of this union type and identifiers of alternatives of the union type containing the default alternative shall be distinct.

c) The @default alternative shall not be of the anytype.
EXAMPLE 2:
	type union MyDefaultUnionTypeMyUnionTypeWithDefault
	{
	  @default integer 			number,
	           charstring 		string
	};


	// A valid instantiation of MyDefaultUnionType MyUnionTypeWithDefault would be 
	var MyDefaultUnionType MyUnionTypeWithDefault v_age, v_oneYearOlder;

	v_age := 34; 	// implicit usage of the default alternative: the integer type is
					// compatible with the default alternative; this is a shorthand notation
					// for v_age.number := 34 or v_age := { number := 34 }

	v_oneYearOlder := v_age+1;	// implicit selection of the default alternative: the union
								// default alternative is compatible with integer, so that it
								// can be used as an integer expression; this is equivalient to:
								// v_oneYearOlder.number := v_age.number+1;

	type union MyDefaultUnionType2 {
		@default
		MyDefaultUnionType ageInYears,
		integer ageInDays
	}

	var MyDefaultUnionType2 v_age2 := 3;	// nested default usage: 3 is compatible with 
											// both alternatives, but only alternative ageInYears
											// has @default, so this is equivalent to
											// v_age2 := { ageInYears := 3 } which is equivalent

											// to v_age2 := { ageInYears := { number := 3 } }
	var integer v_result := v_age + v_age2;	// v_result is 37 as the expression is equivalent
												// to v_age.number + v_age2.ageInYears
	v_age := {string := "I feel young"};
	v_result := v_age + v_age2;			// test case error: v_age would be treated as
											// v_age.number, which is not the selected alternative

	type union MyUnionTypeWithDefaultErr
	{
	  @default MyDefaultUnionType ageInYears,
	           charstring 		  string	// produces an error as the identifier "string" is
	};										// already used in the default alternative

Alternatives of a union type shall be referenced by the dot notation TypeIdOrExpression.AlternativeId, where TypeIdOrExpression resolves to the name of a union type or an expression of a union type such as variable, formal parameter, module parameter, constant, template, or function invocation. AlternativeId shall resolve to the name of an alternative in the union type or in case of an anytype value or template AlternativeId shall resolve to a known type name or a known type name qualified with a module name. Alternatives of union type definitions shall not reference themselves.
EXAMPLE 1:
	v_myVar5 := v_myUnion1.myChoice1;
	// If a union type is nested in another type then the reference may look like this 
	v_myVar6 := v_myRecord1.myElement1.myChoice2;
	// Note, that the union type, of which the field with the identifier 'myChoice2' is referenced,
	// is embedded in a record type

If an alternative in a union type or a subtype of a union type is referenced by the dot notation, the resulting type is the set of values allowed for that alternative imposed by the constraints of the alternative declaration itself (i.e. any constraints applied to the union type itself are ignored).
When an alternative of a union type is referenced on the right hand side of an assignment an error shall occur if the referenced alternative is not the currently chosen alternative or if the referenced union field or value is omitted or uninitialized.
EXAMPLE 2:
	type union MyUnion2
	{
		integer choice1,
		charstring choice2
	}
	type record MyRecordEmbedsUnion
	{
		MyUnion2 field1 optional
	}
	...
	var MyUnion2 v_un2 := { choice1 := 1 }
	var charstring v_char := v_un2.choice2; // causes an error as v_un.choice2 is not chosen
	var MyRecordEmbedsUnion v_rec := { field1 := omit }
	var integer v_int := v_rec.field1.choice1; // causes an error as v_rec.field1 is omitted

When referencing an alternative of a union type on the left hand side of an assignment, the referenced alternative shall become the chosen one. This rule shall apply recursively if the reference contains alternatives of nested unions, choosing all the referenced alternatives.
When referencing an alternative of an uninitialized union value or field or omitted field (including omitting a field at a higher level of the embedding hierarchy) on the left hand side of an assignment, the reference shall recursively be expanded up to and including the depth of the referenced alternative as follows:
1. When expanding a value or value field of union type, the alternative referenced in the dot notation becomes the chosen one.
Expansion of record, record of, set, set of, and array values and intermediate fields shall follow the rules of item a) in clauses 6.2.1.1 and 6.2.3, and clause 6.2.2.1 correspondingly.
At the end of the expansion, the value at the right hand side of the assignment shall be assigned to the referenced alternative.
EXAMPLE 3:
	type union MyUnion3
	{
		integer choice1,
		union
		{
			bitstring subchoice1,
			charstring subchoice2
		} choice2
	}
	...
	var MyUnion3 v_un3 := { choice1 := 1 };
	var MyRecordEmbedsUnion v_rec2 := { field1 := omit };
    v_un3.choice2.subchoice2 := "Hello!"; 
	// after the assignment v_un3 equals to { choice2 := { subchoice2 := "Hello!" } }
	v_rec2.field1.choice1 := 10; // after the assignment v_rec2 equals to
	                             // { field1 := { choice1 := 10 } }

Optional fields are not allowed for the union type, which means that the optional keyword shall not be used with union types.

TTCN‑3 supports the definition of types for union alternatives nested within the union definition, similar to the mechanism for record types described in clause 6.2.1.3.

